

Knowledge Representation & Reasoning

Project 1 - Resolution & Davis-Putnam SAT Solver

Radu-Tudor Vrînceanu, 407 (Artificial Intelligence)

December 9, 2024

University of Bucharest
Faculty of Mathematics and Computer Science
2024-2025



1 Introduction

In the beginning, this project was built using the SWI-Prolog language and Python for the web application using StreamLit as the foundational library that empowers it. In order to start the project and get it up and running please assure that on the machine there is Python v3.10+ installed and also the StreamLit utility is installed using `pip/pip3 install streamlit`. Right after that run the command `streamlit run app.py` and the application will start without any problems on your localhost.

The **Knowledge Base** that needed to be defined is the following one which was inspired from *professor Gordon S. Novak Jr.* - Resolution Algorithm:

1. Every adult loves to play with every lego set.
2. Anyone who plays with some lego set is not a technology fanatic.
3. Anyone who uses a smartphone is a technology fanatic.
4. Anyone who buys a smartphone either craves it or uses it.
5. Paul buys a smartphone.
6. Darth Vader's Scene is one of the lego sets.
7. **Question:** If Paul is an adult, then Paul craves to have a smartphone.

We will use the following symbols to denote the relationships:

1. $Adult(X)$ - denotes if X is an adult;
2. $LEGOSet(Y)$ - denotes if Y is a lego set;
3. $UsesSmartphone(X)$ - denotes if X uses a smartphone;
4. $TechFanatic(X)$ - denotes if X is a technology fanatic;
5. $CravesSmartphone(X)$ - denotes if X craves or not for a smartphone;
6. $BuysSmartphone(X)$ - denotes if X buys a smartphone;
7. $PlaysWith(X, Y)$ - denotes the relationship between X and Y ;
8. DVS - denotes the constant for the Darth Vader's Scene inside the **Knowledge Base**;

9. *Paul* - denotes the constant for Paul inside the **Knowledge Base**.

We will convert the **Knowledge Base** in **First Order Logic** based on the symbols denoted previously:

1.

$$\forall x \forall y (Adult(x) \wedge LEGOSet(y) \Rightarrow PlaysWith(x, y))$$

2.

$$\forall x (\exists y (LEGOSet(y) \wedge PlaysWith(x, y)) \Rightarrow \neg TechFanatic(x))$$

3.

$$\forall x (UsesSmartphone(x) \Rightarrow TechFanatic(x))$$

4.

$$\forall x (BuysSmartphone(x) \Rightarrow (CravesSmartphone(x) \vee UsesSmartphone(x)))$$

5.

$$BuysSmartphone(Paul)$$

6.

$$LEGOSet(DVS)$$

7. **Question:**

$$Adult(Paul) \Rightarrow CravesSmartphone(Paul)$$

We will convert the **First Order Logic** to **CNF**:

$$1. \forall x \forall y (\neg Adult(x) \vee \neg LEGOSet(y) \vee PlaysWith(x, y))$$

$$2. \forall x \forall y (\neg LEGOSet(y) \vee \neg PlaysWith(x, y) \vee \neg TechFanatic(x))$$

$$3. \forall x (\neg UsesSmartphone(x) \vee TechFanatic(x))$$

$$4. \forall x (\neg BuysSmartphone(x) \vee CravesSmartphone(x) \vee UsesSmartphone(x))$$

$$5. \text{ BuysSmartphone}(\text{Paul})$$

$$6. \text{ LEGOSet}(\text{DVS})$$

$$7. \neg \text{Adult}(\text{Paul}) \vee \text{CravesSmartphone}(\text{Paul})$$

To ensure the question is entitled for this **Knowledge Base** we will use the negation of the question stated i.e. $\text{Adult}(\text{Paul}) \wedge \neg \text{CravesSmartphone}(\text{Paul})$. And check if the KB with the negation of the question is UNSATISFIABLE. This means our CNFs will be the following ones:

$$1. \forall x \forall y (\neg \text{Adult}(x) \vee \neg \text{LEGOSet}(y) \vee \text{PlaysWith}(x, y))$$

$$2. \forall x \forall y (\neg \text{LEGOSet}(y) \vee \neg \text{PlaysWith}(x, y) \vee \neg \text{TechFanatic}(x))$$

$$3. \forall x (\neg \text{UsesSmartphone}(x) \vee \text{TechFanatic}(x))$$

$$4. \forall x (\neg \text{BuysSmartphone}(x) \vee \text{CravesSmartphone}(x) \vee \text{UsesSmartphone}(x))$$


$$5. \text{ BuysSmartphone}(\text{Paul})$$

$$6. \text{ LEGOSet}(\text{DVS})$$

$$7. \text{ Adult}(\text{Paul})$$

$$8. \neg \text{CravesSmartphone}(\text{Paul})$$

The manual demonstration of this Knowledge base is the following one:



2 Resolution

The resolution algorithm is a refutation-based proof technique used to derive a contradiction from a set of logical clauses, thereby proving unsatisfiability. It operates on clauses in Conjunctive Normal Form (CNF) and relies on the refutation strategy, meaning that the algorithm assumes the negation of the statement to be proved and tries to derive a contradiction. The code for the resolution procedure is the following one:

```

1 read_input(FilePath, X) :-
2     see(FilePath), read(X), seen.
3
4 retrieve_all_literals_selection(KB, Lit) :-
5     member(Clause, KB),
6     member(Lit, Clause).
7 retrieve_negation_literals_selection(AllLiterals, NegLit) :-
8     member(Lit, AllLiterals),
9     negate(Lit, NegLit),
10    \+ member(NegLit, AllLiterals).
11 exclude_pure_clauses_from_kb(KB, NewKB) :-
12     findall(Lit, retrieve_all_literals_selection(KB, Lit), AllLiterals),
13     findall(NegLit, retrieve_negation_literals_selection(AllLiterals, NegLit), NotFoundLiterals),
14     exclude(is_pure_clause(NotFoundLiterals), KB, NewKB).
15 is_pure_clause(NotFoundLiterals, Clause) :-
16     member(Lit, Clause), negate(Lit, NegLit), member(NegLit, NotFoundLiterals).
17
18 exclude_tautologies_from_kb(KB, NewKB) :- exclude(is_tautology, KB, NewKB).

```

```

19 is_tautology(Clause) :- member(Lit, Clause), negate(Lit, NegLit), member(NegLit, Clause).
20
21 exclude_subsumed_clauses_from_kb(KB, NewKB) :- exclude(is_subsummed(KB), KB, NewKB).
22 is_subsummed(KB, Clause) :- member(NewClause, KB), NewClause \= Clause, subset(NewClause, Clause).
23
24 negate(n(X), X) :- !.
25 negate(X, n(X)).
26
27 solve(FirstClause, SecondClause, Resultant) :-
28     member(Lit, FirstClause),
29     negate(Lit, NegLit),
30     member(NegLit, SecondClause),
31     select(Lit, FirstClause, RestOfFirstClause),
32     select(NegLit, SecondClause, RestOfSecondClause),
33     append(RestOfFirstClause, RestOfSecondClause, IntermediateResultant),
34     sort(IntermediateResultant, Resultant).
35
36 resolve_kb(KB, NewKB) :-
37     findall(Resultant,
38         (member(FirstClause, KB),
39          member(SecondClause, KB),
40          FirstClause @< SecondClause,
41          solve(FirstClause, SecondClause, Resultant),
42          \+ member(Resultant, KB)),
43         NewClauses),
44     append(KB, NewClauses, TempKB),
45     sort(TempKB, NewKB).
46
47 apply_solve_resolution(KB) :-
48     (member([], KB) => write('UNSATISFIABLE'), nl, ! ;
49      resolve_kb(KB, NewKB),
50      (KB == NewKB => write('SATISFIABLE'), nl ; apply_solve_resolution(NewKB))).
51
52 resolution_fol(FilePath) :-
53     read_input(FilePath, KB),
54     write('Initial_KB: '), write(KB), nl,
55     apply_solve_resolution(KB).
56
57 resolution_propositional(FilePath) :-
58     read_input(FilePath, KB),
59     write('Initial_KB: '), write(KB), nl,
60     exclude_pure_clauses_from_kb(KB, FilteredPureClausesKB),
61     write('Our_KB_after_pure_classes_removal: '), write(FilteredPureClausesKB), nl,
62     exclude_tautologies_from_kb(FilteredPureClausesKB, FilteredTautologiesClausesKB),
63     write('Our_KB_after_tautologies_removal: '), write(FilteredTautologiesClausesKB), nl,
64     exclude_subsumed_clauses_from_kb(FilteredTautologiesClausesKB, NewKB),
65     write('Our_KB_after_subsumed_clauses_removal: '), write(NewKB), nl,
66     apply_solve_resolution(KB).

```

3 Davis-Putnam SAT Solver

The Davis-Putnam (DP) algorithm is a classic method for solving the Boolean satisfiability problem (SAT), which involves determining if there exists an assignment of truth values to variables that makes a given Boolean formula true. The algorithm, introduced by Martin Davis and Hugh Putnam in 1960, is a backtracking-based approach that recursively eliminates variables and their associated clauses.

Strategies to select the atom for the dot operation that were implemented for this project were:

1. the **most frequent** atom present in all the clauses;
2. the atom in the **shortest clause**.

```

1 retrieve_p_by_value(P, ValueOfP, NewP) :-
2 (
3     ValueOfP = true -> NewP = P
4     ;
5     ValueOfP = false -> negate(P, NegP), NewP = NegP
6 ).
7
8 dot_operation([], _, _, []) :- !.
9 dot_operation([CurrentClause | RestClauses], P, ValueOfP, NewClauses) :-
10     retrieve_p_by_value(P, ValueOfP, NewP),
11     (member(NewP, CurrentClause) ->
12         dot_operation(RestClauses, P, ValueOfP, NewClauses)
13         ;
14         negate(NewP, NegP), (member(NegP, CurrentClause) ->
15             delete(CurrentClause, NegP, NewCurrentClause),
16             (NewCurrentClause == [] -> fail ; true),
17             dot_operation(RestClauses, P, ValueOfP, RecursiveNewClauses),
18
19             NewClauses = [NewCurrentClause | RecursiveNewClauses]
20             ;
21             dot_operation(RestClauses, P, ValueOfP, RecursiveNewClauses),
22
23             NewClauses = [CurrentClause | RecursiveNewClauses]
24         )
25     ).
26
27 choose_most_frequent_atom(KB, P) :-
28     findall(Lit, retrieve_all_literals_selection(KB, Lit), AllLiterals),
29     construct_frequency_list(AllLiterals, P).
30 % https://stackoverflow.com/questions/50437617/prolog-function-that-returns-the-most-frequent-element-in-a-list
31 construct_frequency_list(AllLiterals, P) :-
32     sort(AllLiterals, UniqueLiterals),
33     findall([Freq, X], (member(X, UniqueLiterals), include(=(X), AllLiterals, XX), length(XX, Freq)), Freqs),
34     sort(Freqs, SFreqs),
35     last(SFreqs, [Freq, P]).
36 apply_solve_davis_putnam_most_frequent([], []) :- !.
37 apply_solve_davis_putnam_most_frequent(KB, _) :- member([], KB), !, fail.
38 apply_solve_davis_putnam_most_frequent(KB, Solution) :-
39     choose_most_frequent_atom(KB, P),
40     (
41         dot_operation(KB, P, true, NewKBTrue),
42         write('ChosenAtom: '), write(P), write('_with_value_TRUE'), nl,
43         write('Clauses_after_dot_operation: '), write(NewKBTrue), nl,
44         apply_solve_davis_putnam_most_frequent(NewKBTrue, RecursiveSolutions),
45
46         Solution = [atom(P, true) | RecursiveSolutions]
47         ;
48         dot_operation(KB, P, false, NewKBFalse),
49         write('ChosenAtom: '), write(P), write('_with_value_FALSE'), nl,
50         write('Clauses_after_dot_operation: '), write(NewKBFalse), nl,
51         apply_solve_davis_putnam_most_frequent(NewKBFalse, RecursiveSolutions),
52
53         Solution = [atom(P, false) | RecursiveSolutions]
54     ).
55 davis_putnam_most_frequent(FilePath) :-
56     read_input(FilePath, KB),
57     write('Initial_KB: '), write(KB), nl,
58     (apply_solve_davis_putnam_most_frequent(KB, Solution) ->
59         write('YES'), nl, write('Solution: '), write(Solution)
60         ;
61         write('NO')
62     ).
63
64 choose_shortest_clause_atom(KB, P) :-

```

```

65     findall([Length, Clause], (member(Clause, KB), length(Clause, Length)), AllClauseLengths),
66     sort(AllClauseLengths, SortedAllClauseLengths),
67     SortedAllClauseLengths = [[_, ShortestClause] | _],
68     member(P, ShortestClause).
69 apply_solve_davis_putnam_shortest_clause([], []) :- !.
70 apply_solve_davis_putnam_shortest_clause(KB, _) :- member([], KB), !, fail.
71 apply_solve_davis_putnam_shortest_clause(KB, Solution) :-
72     choose_shortest_clause_atom(KB, P),
73     (
74         dot_operation(KB, P, true, NewKBTrue),
75         write('ChosenAtom: '), write(P), write('_with_value_TRUE'), nl,
76         write('Clauses_after_dot_operation: '), write(NewKBTrue), nl,
77         apply_solve_davis_putnam_most_frequent(NewKBTrue, RecursiveSolutions),
78
79         Solution = [atom(P, true) | RecursiveSolutions]
80     ;
81         dot_operation(KB, P, false, NewKBFalse),
82         write('ChosenAtom: '), write(P), write('_with_value_FALSE'), nl,
83         write('Clauses_after_dot_operation: '), write(NewKBFalse), nl,
84         apply_solve_davis_putnam_most_frequent(NewKBFalse, RecursiveSolutions),
85
86         Solution = [atom(P, false) | RecursiveSolutions]
87     ).
88
89 davis_putnam_shortest_clause(FilePath) :-
90     read_input(FilePath, KB),
91     write('Initial_KB: '), write(KB), nl,
92     (apply_solve_davis_putnam_shortest_clause(KB, Solution) ->
93         write('YES'), nl, write('Solution: '), write(Solution)
94     ;
95         write('NO')
96     ).

```